
RALFit computes a solution 𝑥 to the non-linear least-squares problem

min
𝑥

𝐹 (𝑥) :=
1

2
‖𝑟(𝑥)‖2𝑊 +

𝜎

𝑝
‖𝑥‖𝑝2,

where 𝑊 ∈ R𝑚×𝑚 is a diagonal, non-negative, weighting matrix, and 𝑟(𝑥) = (𝑟1(𝑥), 𝑟2(𝑥), ..., 𝑟𝑚(𝑥))𝑇 is a
non-linear function.

A typical use may be to fit a function 𝑓(𝑥) to the data 𝑦𝑖, 𝑡𝑖, weighted by the uncertainty of the data, 𝜎𝑖, so that

𝑟𝑖(𝑥) := 𝑦𝑖 − 𝑓(𝑥; 𝑡𝑖),

and 𝑊 is the diagonal matrix such that 𝑊𝑖𝑖 = (1/
√
𝜎𝑖). For this reason we refer to the function 𝑟 as the residual

function.

Various algorithms for solving this problem are implemented – see Description of the method used.

2 Contents



CHAPTER 1

Contents

1.1 Installation

1.1.1 Obtaining the code

The latest version of the code can be downloaded from GitHub by issuing the command

git clone https://github.com/ralna/RALFit.git

1.1.2 Building the library

From the RALFit/libRALFit/ directory, issue the commands:

mkdir build
cd build
cmake ..
make

1.2 How to use the package

1.2.1 Overview

Calling sequences

Access to the package requires a USE statement

use ral_nlls_double

3

The user can then call one of the procedures:

nlls_solve() solves the non-linear least squares problem.

nlls_iterate() performs one iteration of the non-linear least squares solver.

The calling sequences of these subroutines are outlined in Argument lists and calling sequences.

The derived data types

For each problem, the user must employ the derived types defined by the module to declare scalars of the
types nlls_inform and nlls_options. If nlls_iterate() is to be used, then a scalar of the type
nlls_workspace must also be defined. The following pseudocode illustrates this.

use nlls_module
!...
type (NLLS_inform) :: inform
type (NLLS_options) :: options
type (NLLS_workspace) :: work ! needed if nlls_iterate to be called
!...

The components of nlls_options and nlls_inform are explained below in Data types.

1.2.2 Argument lists and calling sequences

We use square brackets to indicate arguments. In each call, optional arguments follow the argument inform. Since we
reserve the right to add additional optional arguments in future releases of the code, we strongly recommend that all
optional arguments be called by keyword, not by position.

The term package type is used to mean double precision.

To solve the non-linear least squares problem

subroutine nlls_solve(n, m, X, eval_r, eval_J, eval_Hf, params, options, inform[, weights, eval_HP
])

Solves the non-linear least squares problem.

Parameters

• n [integer,in] :: holds the number 𝑛 of variables to be fitted; i.e., 𝑛 is the length of the
unknown vector 𝑥. Restriction: n > 0.

• m [integer,in] :: holds the number 𝑚 of data points available; i.e., 𝑚 is the number of
residuals 𝑟𝑖. Restriction: m ≥ 0.

• X (n) [real,inout] :: on entry, it must hold the initial guess for 𝑥, and on successful exit it
holds the solution to the non-linear least squares problem.

• eval_r [procedure] :: given a point 𝑥𝑘, returns the vector 𝑟(𝑥𝑘). Further details of the
format required are given in eval_r() in User-supplied function evaluation routines.

• eval_J [procedure] :: given a point 𝑥𝑘, returns the 𝑚×𝑛 Jacobian matrix, 𝐽𝑘, of 𝑟 evaluated
at 𝑥𝑘. Further details of the format required are given in eval_J() in User-supplied
function evaluation routines.

• eval_Hf [procedure] :: given vectors 𝑥 ∈ R𝑛 and 𝑟 ∈ R𝑚, returns the quantity∑︀𝑚
𝑖=1(𝑟)𝑖∇2𝑟𝑖(𝑥). Further details of the format required are given in eval_Hf()

in User-supplied function evaluation routines. If exact_second_derivative = .
false. in nlls_options, then this is not referenced.

• params [params_base_type,in] :: holds parameters to be passed to the user-defined routines
eval_r(), eval_J(), and eval_Hf(). Further details of its use are given in User-
supplied function evaluation routines.

• options [nlls_options,in] :: controls execution of algorithm.

• inform [nlls_inform,out] :: components provide information about the execution of the
subroutine.

Options

• weights (n) [real] :: If present, this holds the square-roots of the diagonal entries of the
weighting matrix, 𝑊 . If absent, then the norm in the least squares problem is taken to be
the 2-norm, that is, 𝑊 = 𝐼 .

• eval_HP [procedure] :: If present, this is a routine that, given vectors 𝑥,𝑦 ∈ R𝑚, returns
the matrix 𝑃 (𝑥,𝑦) := (𝐻1(𝑥)𝑦 . . . 𝐻𝑚(𝑥)𝑦). Further details of the format required are
given in eval_HP() in User-supplied function evaluation routines. This is only referenced
if model = 4 in nlls_options.

To iterate once

subroutine nlls_iterate(n, m, X, eval_r, eval_J, eval_Hf, params, options, inform[, weights])
A call of this form allows the user to step through the solution process one iteration at a time.

n, m, eval_F, eval_J, eval_HF, params, info, options and weights are as in the desciption of nlls_solve().

Parameters

• X (n) [real,inout] :: on the first call it must hold the initial guess for 𝑥. On return it holds
the value of 𝑥 at the current iterate, and must be passed unaltered to any subsequent call to
nlls_iterate().

• w [nlls_workspace,inout] :: is used to store the current state of the iteration and should not
be altered by the user.

The user may use the components convergence_normf and convergence_normg and converge_norms
in nlls_inform to determine whether the iteration has converged.

1.2.3 User-supplied function evaluation routines

The user must supply routines to evaluate the residual, Jacobian and Hessian at a point. RALFit will call these routines
internally.

In order to pass user-defined data into the evaluation calls, params_base_type is extended to a user_type, as
follows:

type, extends( params_base_type ) :: user_type
! code declaring components of user_type

end type user_type

We recommend this type is wrapped in a module with the user-defined routines for evaluating the function, Jacobian,
and Hessian.

The components of the extended type are accessed through a select type construct:

1.2. How to use the package 5

select type(params)
type is(user_type)

! code that accesses components of params that were defined within user_type
end select

For evaluating the function 𝑟(𝑥)

A subroutine must be supplied to calculate 𝑟(𝑥) for a given vector 𝑥. It must implement the following interface:

abstract interface
subroutine eval_r(status, n, m, x, r, params)

integer, intent(inout) :: status
integer, intent(in) :: n
integer, intent(in) :: m
double precision, dimension(n), intent(in) :: x
double precision, dimension(m), intent(out) :: r
class(params_base_type), intent(in) :: params

end subroutine eval_r
end interface

subroutine eval_r(status, n, m, x, r, params)

Parameters

• status [integer,inout] :: is initialised to 0 before the routine is called. If it is set to a non-zero
value by the routine, then nlls_solve() / nlls_iterate() will exit with error.

• n [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• m [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• X (n) [real,in] :: holds the current point 𝑥𝑘 at which to evaluate 𝑟(𝑥𝑘).

• r (m) [real,out] :: must be set by the routine to hold the residual function evaluated at the
current point 𝑥𝑘, 𝑟(𝑥𝑘).

• params [params_base_type,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

For evaluating the function 𝐽 = ∇𝑟(𝑥)

A subroutine must be supplied to calculate 𝐽 = ∇𝑟(𝑥) for a given vector 𝑥. It must implement the following interface:

abstract interface
subroutine eval_J(status, n, m, x, J, params)

integer, intent(inout) :: status
integer, intent(in) :: n
integer, intent(in) :: m
double precision, dimension(n), intent(in) :: x
double precision, dimension(n*m), intent(out) :: J
class(params_base_type), intent(in) :: params

end subroutine eval_J
end interface

subroutine eval_J(status, n, m, x, J, params)

6 Chapter 1. Contents

Parameters

• status [integer,inout] :: is initialised to 0 before the routine is called. If it is set to a non-zero
value by the routine, then nlls_solve() / nlls_iterate() will exit with error.

• n [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• m [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• X (n) [real,in] :: holds the current point 𝑥𝑘 at which to evaluate 𝐽(𝑥𝑘).

• J (m*n) [real,out] :: must be set by the routine to hold the Jacobian of the residual function
evaluated at the current point 𝑥𝑘, 𝑟(𝑥𝑘). J(i*m+j) must be set to hold ∇𝑥𝑗

𝑟𝑖(𝑥𝑘).

• params [params_base_type,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

For evaluating the function 𝐻𝑓 =
∑︀𝑚

𝑖=1 𝑟𝑖(𝑥)𝑊∇2𝑟𝑖(𝑥)

A subroutine must be supplied to calculate 𝐻𝑓 =
∑︀𝑚

𝑖=1(𝑟)𝑖∇2𝑟𝑖(𝑥) for given vectors 𝑥 ∈ R𝑛 and 𝑟 ∈ R𝑚; here
(𝑟)𝑖 denotes the 𝑖th component of the vector 𝑟. The subroutine must implement the following interface:

abstract interface
subroutine eval_Hf_type(status, n, m, x, r, Hf, params)

integer, intent(inout) :: status
integer, intent(in) :: n
integer, intent(in) :: m
double precision, dimension(n), intent(in) :: x
double precision, dimension(m), intent(in) :: r
double precision, dimension(n*n), intent(out) :: Hf
class(params_base_type), intent(in) :: params

end subroutine eval_Hf_type
end interface
:language: fortran

subroutine eval_Hf(status, n, m, x, r, Hf, params)

Parameters

• status [integer,inout] :: is initialised to 0 before the routine is called. If it is set to a non-zero
value by the routine, then nlls_solve() / nlls_iterate() will exit with error.

• n [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• m [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• X (n) [real,in] :: holds the current point 𝑥𝑘 at which to evaluate
∑︀𝑚

𝑖=1(𝑟)𝑖∇2𝑟𝑖(𝑥).

• r (m) [real,in] :: holds 𝑊𝑟(𝑥), the (weighted) residual, as computed from vector returned
by the last call to eval_r().

• Hf (n*n) [real,out] :: must be set by the routine to hold the matrix
∑︀𝑚

𝑖=1(𝑟)𝑖∇2𝑟𝑖(𝑥𝑘), held
by columns as a vector, where (𝑟)𝑖 denotes the 𝑖th component of r, the vector passed to the
routine.

• params [params_base_type,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

1.2. How to use the package 7

For evaluating the function 𝑃 (𝑥,𝑦) := (𝐻1(𝑥)𝑦 . . . 𝐻𝑚(𝑥)𝑦)

A subroutine may be supplied to calculate 𝑃 (𝑥,𝑦) := (𝐻1(𝑥)𝑦 . . . 𝐻𝑚(𝑥)𝑦) for given vectors 𝑥,𝑦 ∈ R𝑛. The
subroutine must implement the following interface:

abstract interface
subroutine eval_HP_type(status, n, m, x, y, HP, params)

integer, intent(inout) :: status
integer, intent(in) :: n
integer, intent(in) :: m
double precision, dimension(n), intent(in) :: x
double precision, dimension(n), intent(in) :: y
double precision, dimension(n*m), intent(out) :: HP
class(params_base_type), intent(in) :: params

end subroutine eval_HP_type
end interface
:language: fortran

subroutine eval_HP(status, n, m, x, y, HP, params)

Parameters

• status [integer,inout] :: is initialised to 0 before the routine is called. If it is set to a non-zero
value by the routine, then nlls_solve() / nlls_iterate() will exit with error.

• n [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• m [integer,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

• x (n) [real,in] :: holds the current point 𝑥𝑘 at which to evaluate the Hessians∇2𝑟𝑖(𝑥𝑘).

• y (n) [real,in] :: holds 𝑦, the vector which multiplies each Hessian.

• HP (n*m) [real,out] :: must be set by the routine to hold the matrix 𝑃 (𝑥,𝑦) :=
(𝐻1(𝑥)𝑦 . . . 𝐻𝑚(𝑥)𝑦), held by columns as a vector.

• params [params_base_type,in] :: is passed unchanged as provided in the call to
nlls_solve()/nlls_iterate().

1.2.4 Data types

The derived data type for holding options

type nlls_options
This is used to hold controlling data. The components are automatically given default values in the definition of
the type.

Printing Controls

Type fields

• % error [integer,default=6] :: the Fortran unit number for error messages. If it is nega-
tive, these messages will be suppressed.

• % out [integer,default=6] :: the Fortran unit number for general messages. If it is negative,
these messages will be suppressed.

• % print_level [integer,default=0] :: controls the level of output required. Options are:

< 1 No informational output will occur.
1 Gives a one-line summary for each iteration.
2 As 1, plus gives a summary of the inner iteration for each iteration.
3 As 2, plus gives more verbose (debugging) output.

Choice of Algorithm

Type fields

• % model [integer,default=3] :: specifies the model, 𝑚𝑘(·), used. Possible values are:

1 Gauss-Newton (no Hessian).
2 (Quasi-)Newton (uses exact Hessian if exact_second_derivatives is true,

otherwise builds an approximation to the Hessian).
3 Hybrid method (mixture of Gauss-Newton/(Quasi-)Newton, as determined by the

package).
4 Newton-tensor method.

See The models for details.

• % type_of_method [integer,default=1] :: specifies the type of globalization method
used. Possible values are:

1 Trust-region method.
2 Regularization.

See Subproblem solves for details.

• % nlls_method [integer,default=4] :: specifies the method used to solve (or approxi-
mate the solution to) the trust-region sub problem. Possible values are:

1 Powell’s dogleg method (approximates the solution).
2 The Adachi-Iwata-Nakatsukasa-Takeda (AINT) method.
3 The More-Sorensen method.
4 Galahad’s DTRSmethod if type_of_method=1, or Galahad’s DRQSmethod

if type_of_method=2.

See Subproblem solves for details.

• % exact_second_derivatives [logical,default=false] :: if true, signifies that the
exact second derivatives are available (and, if false, approximates them using a secant
method).

Solving a regularized problem

Type fields

• % regularization [integer,default=0] :: specifies if a regularized non-linear least
squares problem needs to be solved, and if so, what method is used to solve it. Possible
values are:

1.2. How to use the package 9

0 𝜎 = 0, and an unregularized problem is solved
1 a non-linear least-squares problem of size 𝑛 + 𝑚 is solved implicitly. Can only be

used if 𝑝 = 2.
2 a non-linear least-squares problem of size 𝑛 + 1 is solved implicitly.

See Incorporating the regularization term for details.

• % regularization_term [real,default=0.0] :: specifies the regularization weight, 𝜎,
used in the least-squares problem.

• % regularization_power [real,default=0.0] :: specifies the regularization index, 𝑝,
used in the least-squares problem.

Stopping rules

Type fields

• % maxit [integer,default=100] :: gives the number of iterations the algorithm is allowed
to take before being stopped. This is not accessed if nlls_iterate() is used.

• % stop_g_absolute [real,default=1e-5] :: specifies the absolute tolerance used in the
convergence test on ‖𝐽𝑘

𝑇𝑟(𝑥𝑘))‖/‖𝑟(𝑥𝑘)‖.

• % stop_g_relative [real,default=1e-8] :: specifies the relative tolerance used in the
convergence test on ‖𝐽𝑘

𝑇𝑟(𝑥𝑘)‖/‖𝑟(𝑥𝑘)‖.

• % stop_f_absolute [real,default=1e-5] :: specifies the absolute tolerance used in the
convergence test on ‖𝑟(𝑥𝑘)‖.

• % stop_f_relative [real,default=1e-8] :: specifies the relative tolerance used in the
convergence test on ‖𝑟(𝑥𝑘)‖.

• % stop_s [real,default=eps] :: specifies the tolerance used in the convergence test on
‖𝑠𝑘‖.

Trust region radius/regularization behaviour

Type fields

• % relative_tr_radius [integer,default=0] :: specifies whether the initial trust region
radius should be scaled.

• % initial_radius_scale [real,default=1.0] :: specifies the scaling parameter for
the initial trust region radius, which is only used if relative_tr_radius = 1.

• % initial_radius [real,default=100.0] :: specifies the initial trust-region radius, ∆.

• % maximum_radius [real,default=1e8] :: specifies the maximum size permitted for the
trust-region radius.

• % eta_successful [real,default=1e-8] :: specifies the smallest value of 𝜌 such that the
step is accepted. .. success_but_reduce is also available, but not documented

• % eta_very_successful [real,default=0.9] :: specifies the value of 𝜌 after which the
trust-region radius is increased.

• % eta_too_successful [real,default=2.0] :: specifies that value of 𝜌 after which the
step is accepted, but keep the trust-region radius unchanged.

• % radius_increase [real,default=2.0] :: specifies the factor to increase the trust-
region radius by.

10 Chapter 1. Contents

• % radius_reduce [real,default=0.5] :: specifies the factor to decrease the trust-region
radius by.

• % tr_update_strategy [integer,default=1] :: specifies the strategy used to update
∆𝑘. Possible values are:

1 use the usual step function.
2 use a the continuous method.

Scaling options

Type fields

• % scale [integer,default=1] :: specifies how, if at all, we scale the Jacobian. We calculate
a diagonal scaling matrix, D, as follows:

• % scale_trim_max [logical,default=true] :: specifies whether or not to trim large val-
ues of the scaling matrix, 𝐷. If true, D𝑖,𝑖 ← 𝑚𝑖𝑛(D𝑖,𝑖, scale_max).

• % scale_max [real,default=1e11] :: specifies the maximum value allowed if
scale_trim_max = true.

• % scale_trim_min [logical,default=true] :: specifies whether or not to trim small val-
ues of the scaling matrix, D. If true, D𝑖,𝑖 ← 𝑚𝑎𝑥(D𝑖,𝑖, scalemax).

• % scale_min [real,default=1e-11] :: specifies the minimum value allowed if
scale_trim_max = true.

• % scale_require_increase [logical,default=false] :: specifies whether or not to
require D𝑖,𝑖 to increase before updating it.

Hybrid method options These options are used if model=3

Type fields

• % hybrid_switch [real,default=0.1] :: specifies the value, if model=3, at which sec-
ond derivatives are used.

• % hybrid_tol [real,default=2.0] :: if model=3, specifies the value such that if
‖𝐽𝑘

𝑇𝑊𝑟(𝑥𝑘)‖2 < hybrid_tol * 0.5‖𝑟(𝑥𝑘)‖2𝑊 the method switches to a quasi-Newton
method.

• % hybrid_switch_its [integer,default=1] :: if model=3, sets how many iterates in
a row must the condition in the definition of hybrid_tol hold before a switch.

Newton-Tensor options These options are used if model=4

Type fields

• % reg_order [real,default=0.0] :: if nlls_method = 4, the order of the regulariza-
tion used (𝑝 in TODO (eq:: reg_subproblem)). If reg_order = 0.0, then the algorithm
chooses an appropriate value of 𝑝.

• % inner_method [integer,default=2] :: if nlls_method = 4, specifies the method
used to pass in the regularization parameter to the inner non-linear least squares solver.
Possible values are:

1 The current regularization parameter is passed in as a base regularization parameter.
2 A larger non-linear least squares problem is explicitly formed to be solved as the

subproblem.
3 The regularization is handled implicitly by calling RALFit recursively.

1.2. How to use the package 11

More-Sorensen options These options are used if nlls_method=3

Type fields

• % more_sorensen_maxits [integer,default=3] :: if nlls_method = 3, specifies
the maximum number of iterations allowed in the More-Sorensen method.

• % more_sorensen_maxits :: if nlls_method = 3, specifies the maximum num-
ber of iterations allowed in the More-Sorensen method.

• % more_sorensen_shift [real,default=1e-13] :: if nlls_method = 3, specifies
the shift to be used in the More-Sorensen method.

• % more_sorensen_tiny [real,default=10.0*eps] :: if nlls_method = 3, speci-
fies the value below which numbers are considered to be essentially zero.

• % more_sorensen_tol [real,default=1e-3] :: if nlls_method = 3, specifies the
tolerance to be used in the More-Sorensen method.

Other options

Type fields

• % calculate_svd_J [logical,default=false] :: specifies whether or not to calculate the
singular value decomposition of J at each iteration.

• % output_progress_vectors [logical,default=false] :: if true, outputs the progress
vectors nlls_inform%resvec and nlls_inform%gradvec at the end of the rou-
tine.

The derived data type for holding information

type nlls_inform
This is used to hold information about the progress of the algorithm.

Type fields

• % status [integer] :: gives the exit status of the subroutine. See Warning and error
messages for details.

• % error_message (80) [character] :: holds the error message corresponding to the exit
status.

• % alloc_status [integer] :: gives the status of the last attempted alloca-
tion/deallocation.

• % bad_alloc (80) [character] :: holds the name of the array that was being allocated
when an error was flagged.

• % iter [integer] :: gives the total number of iterations performed.

• % f_eval [integer] :: gives the total number of evaluations of the objective function.

• % g_eval [integer] :: gives the total number of evaluations of the gradient of the objective
function.

• % h_eval [integer] :: gives the total number of evaluations of the Hessian of the objective
function.

• % convergence_normf [integer] :: tells us if the test on the size of 𝑟 is satisfied.

• % convergence_normf :: that tells us if the test on the size of the gradient is satisfied.

• % convergence_normf :: that tells us if the test on the step length is satisfied.

• % resvec (iter+1) [real] :: if output_progress_vectors=true in
nlls_options, holds the vector of residuals.

• % resvec :: if output_progress_vectors=true in nlls_options, holds the
vector of gradients.

• % obj [real] :: holds the value of the objective function at the best estimate of the solution
determined by the algorithm.

• % norm_g [real] :: holds the gradient of the objective function at the best estimate of the
solution determined by the package.

• % scaled_g [real] :: holds the gradient of the objective function at the best estimate of
the solution determined by the package.

• % external_return [integer] :: gives the error code that was returned by a call to an
external routine.

• % external_name (80) [character] :: holds the name of the external code that flagged
an error.

The workspace derived data type

type nlls_workspace
This is used to save the state of the algorithm in between calls to nlls_iterate(), and must be used if that
subroutine is required. It’s components are not designed to be accessed by the user.

1.2.5 Warning and error messages

A successful return from a subroutine in the package is indicated by status in nlls_inform having the value
zero. A non-zero value is asscociated with an error message, which will be output on error in nlls_inform.

Possible values are:

-1 Maximum number of iterations reached without convergence.
-2 Error from evaluating a function/Jacobian/Hessian.
-3 Unsupported choice of model.
-4 Error return from an external routine.
-5 Unsupported choice of method.
-6 Allocation error.
-7 Maximum number of reductions of the trust radius reached.
-8 No progress being made in the solution.
-9 n > m.
-10 Unsupported trust region update strategy.
-11 Unable to valid step when solving trust region subproblem.
-12 Unsupported scaling method.
-13 Error accessing pre-allocated workspace.
-14 Unsupported value in type_of_method
-101 Unsupported model in dogleg (nlls_method = 1).
-201 All eigenvalues are imaginary (nlls_method=2).
-202 Matrix with odd number of columns sent to max_eig subroutine (nlls_method=2).
-301 more_sorensen_max_its is exceeded in more_sorensen subroutine (nlls_method=3).
-302 Too many shifts taken in more_sorensen subroutine (nlls_method=3).
-303 No progress being made in more_sorensen subroutine (nlls_method=3).
-401 model = 4 selected, but exact_second_derivatives is set to false.

1.2. How to use the package 13

1.3 Description of the method used

RALFit computes a solution 𝑥 to the non-linear least-squares problem

min
𝑥

𝐹 (𝑥) :=
1

2
‖𝑟(𝑥)‖2𝑊 +

𝜎

𝑝
‖𝑥‖𝑝2, (1.1)

Here we describe the method used to solve (1.1). RALFit implements an iterative method that, at each iteration,
calculates and returns a step 𝑠 that reduces the model by an acceptable amount by solving (or approximating a solution
to) a subproblem, as detailed in Subproblem solves.

The algorithm is iterative. At each point, 𝑥𝑘, the algorithm builds a model of the function at the next step, 𝐹 (𝑥𝑘 + 𝑠𝑘),
which we refer to as 𝑚𝑘(·). We allow either a Gauss-Newton model, a (quasi-)Newton model, or a Newton-tensor
model; see The models for more details.

Once the model has been formed we find a candidate for the next step by solving a subitable subproblem. The quantity

𝜌 =
𝐹 (𝑥𝑘)− 𝐹 (𝑥𝑘 + 𝑠𝑘)

𝑚𝑘(𝑥𝑘)−𝑚𝑘(𝑥𝑘 + 𝑠𝑘)
(1.2)

is then calculated. If this is sufficiently large we accept the step, and 𝑥𝑘+1 is set to 𝑥𝑘 + 𝑠𝑘; if not, the parameter ∆𝑘

is reduced and the resulting new trust-region sub-problem is solved. If the step is very successful – in that 𝜌 is close to
one – ∆𝑘 is increased. Details are explained in Accepting the step and updating the parameter.

This process continues until either the residual, ‖𝑟(𝑥𝑘)‖𝑊 , or a measure of the gradient,
‖𝐽𝑘

𝑇𝑊𝑟(𝑥𝑘)‖2/‖𝑟(𝑥𝑘)‖𝑊 , is sufficiently small.

1.3.1 The models

A vital component of the algorithm is the choice of model employed. There are four choices available, controlled by
the parameter model of nlls_options.

model = 1 this implements the Gauss-Newton model. Here we replace 𝑟(𝑥𝑘 +𝑠) by its first-order Taylor approx-
imation, 𝑟(𝑥𝑘) + 𝐽𝑘𝑠. The model is therefore given by

𝑚𝐺𝑁
𝑘 (𝑠) =

1

2
‖𝑟(𝑥𝑘) + 𝐽𝑘𝑠‖2𝑊 . (1.3)

model = 2 this implements the Newton model. Here, instead of approximating the residual, 𝑟(·), we take as our
model the second-order Taylor approximation of the function, 𝐹 (𝑥𝑘+1). Namely, we use

where 𝑔𝑘 = 𝐽𝑘
𝑇𝑊𝑟(𝑥𝑘) and 𝐻𝑘 =

∑︀𝑚
𝑖=1 𝑟𝑖(𝑥𝑘)𝑊∇2𝑟𝑖(𝑥𝑘). Note that 𝑚𝑁

𝑘 (𝑠) = 𝑚𝐺𝑁
𝑘 (𝑠) + 1

2𝑠
𝑇𝐻𝑘𝑠.

If the second derivatives of 𝑟(·) are not available (i.e., the option exact_second_derivatives is set to
false, then the method approximates the matrix 𝐻𝑘; see Approximating the Hessian.

model = 3 This implements a hybrid model. In practice the Gauss-Newton model tends to work well far away
from the solution, whereas Newton performs better once we are near to the minimum (particularly if the
residual is large at the solution). This option will try to switch between these two models, picking the
model that is most appropriate for the step. In particular, we start using 𝑚𝐺𝑁

𝑘 (·), and switch to 𝑚𝑁
𝑘 (·) if

‖𝑔𝑘‖2 ≤ hybrid_tol 1
2‖𝑟(𝑥𝑘)‖2𝑊 for more than hybrid_switch_its iterations in a row. If, in subse-

quent iterations, we fail to get a decrease in the function value, then the algorithm interprets this as being not
sufficiently close to the solution, and thus switches back to using the Gauss-Newton model.

14 Chapter 1. Contents



RALFit Documentation, Release 0.0.1

The exact method used is described below:

if use_second_derivatives // previous step used Newton model

if‖g𝑘+1‖ > ‖g𝑘‖
use_second_derivatives = false // Switch back to Gauss-Newton

Hf𝑡𝑒𝑚𝑝 = Hf𝑘, Hf𝑘 = 0 // Copy Hessian back to temp array

endif

else

if‖g𝑘+1‖/normF𝑘+1 < hybrid_tol

hybrid_count = hybrid_count + 1 // Update the no of successive failures

if hybrid_count = hybrid_count_switch_its

use_second_derivatives = true

hybrid_count = 0

Hf𝑡𝑒𝑚𝑝 = Hf𝑘// Copy approximate Hessian back

end if
end if

end if

model = 4 this implements a Newton-tensor model. This uses a second order Taylor approximation to the residual,
namely

𝑟𝑖(𝑥𝑘 + 𝑠) ≈ (𝑡𝑘(𝑠))𝑖 := 𝑟𝑖(𝑥𝑘) + (𝐽𝑘)𝑖𝑠 +
1

2
𝑠𝑇𝐵𝑖𝑘𝑠,

where (𝐽𝑘)𝑖 is the ith row of 𝐽𝑘, and 𝐵𝑖𝑘 is∇2𝑟𝑖(𝑥𝑘). We use this to define our model

𝑚𝑁𝑇
𝑘 (𝑠) =

1

2
‖𝑡𝑘(𝑠)‖2𝑊 . (1.4)

1.3.2 Approximating the Hessian

If the exact Hessian is not available, we approximate it using the method of Dennis, Gay, and Welsch4. The method
used is given as follows:

function Hf𝑘+1 = rank_one_update(d, g𝑘, g𝑘+1, r𝑘+1, J𝑘, Hf𝑘)

y = g𝑘 − g𝑘+1̂︀y = J𝑘
𝑇 r𝑘+1 − g𝑘+1̂︂Hf𝑘 = min

(︂
1,
|d𝑇̂︀y|
|d𝑇 Hf𝑘d|

)︂
Hf𝑘

Hf𝑘+1 = ̂︂Hf𝑘 +
(︀
(̂︀y𝑘+1 − Hf𝑘d)𝑇 d

)︀
/y𝑇 d

It is sometimes the case that this approximation becomes corrupted, and the algorithm may not recover from this. To
guard against this, if model = 3 in nlls_options and we are using this approximation to the Hessian in our
(quasi-Newton) model, we test against the Gauss-Newton model if the first step is unsuccessful. If the Gauss-Newton
step would have been successful, we discard the approximate Hessian information, and recompute the step using
Gauss-Newton.

In the case where model=3, the approximation to the Hessian is updated at each step whether or not it is needed for
the current calcuation.

4 Nocedal, J., & Wright, S. (2006). Numerical optimization. Springer Science & Business Media.

1.3.3 Subproblem solves

The main algorithm calls a number of subroutines. The most vital is the subroutine calculate_step, which finds
a step that minimizes the model chosen, subject to a globalization strategy. The algorithm supports the use of two
such strategies: using a trust-region, and regularization. If Gauss-Newton, (quasi-)Newton, or a hybrid method is used
(model = 1,2,3 in nlls_options), then the model function is quadratic, and the methods available to solve
the subproblem are described in The trust region method and Regularization. If the Newton-Tensor model is selected
(model = 4 in nlls_options), then this model is not quadratic, and the methods available are described in
Newton-Tensor subproblem.

Note that, when calculating the step, if the initial regularization parameter 𝜎 in (1.1) is non-zero, then we must modify
J𝑘

𝑇 J𝑘 to take into account the Jacobian of the modified least squares problem being solved. Practically, this amounts
to making the change

J𝑘
𝑇 J𝑘 = J𝑘

𝑇 J𝑘 +

{︃
𝜎𝐼 if 𝑝 = 2
𝜎𝑝
2 ‖𝑥𝑘‖𝑝−4𝑥𝑘𝑥𝑘

𝑇 otherwise
.

The trust region method

If model = 1, 2, or 3, and type_of_method=1, then we solve the subproblem

𝑠𝑘 = arg min
𝑠

𝑚𝑘(𝑠) s.t. ‖𝑠‖𝐵 ≤ ∆𝑘, (1.5)

and we take as our next step the minimum of the model within some radius of the current point. The method used to
solve this is dependent on the control parameter optionsnlls_method. The algorithms called for each of the options are
listed below:

nlls_method = 1 approximates the solution to (1.5) by using Powell’s dogleg method. This takes as the step a
linear combination of the Gauss-Newton step and the steepest descent step, and the method used is described
here:

function dogleg(J, r, Hf, g, Δ)

𝛼 = ‖g‖2/‖J * g‖2

dsd = 𝛼 g

solve dgn = arg min
x
‖Jx− r‖2

if ‖dgn‖ ≤ ∆then
d = dgn

else if ‖𝛼 dsd‖ ≥ ∆

d = (∆/‖dsd‖)dsd
else

d = 𝛼 dsd + 𝛽 (dgn − 𝛼dsd), where 𝛽 is chosen such that ‖d‖ = ∆

end if

nlls_method = 2 solves the trust region subproblem using the trust region solver of Adachi, Iwata, Nakatsukasa,
and Takeda. This reformulates the problem (1.5) as a generalized eigenvalue problem, and solves that. See1 for
more details.

1 Adachi, Satoru and Iwata, Satoru and Nakatsukasa, Yuji and Takeda, Akiko (2015). Solving the trust region subproblem by a generalized
eigenvalue problem. Technical report, Mathematical Engineering, The University of Tokyo.

nlls_method = 3 this solves (1.5) using a variant of the More-Sorensen method. In particular, we implement
Algorithm 7.3.6 in Trust Region Methods by Conn, Gould and Toint2.

nlls_method = 4 this solves (1.5) by first converting the problem into the form

min
𝑝

𝑤𝑇𝑝 +
1

2
𝑝𝑇𝐷𝑝 s.t. ‖𝑝‖ ≤ ∆,

where 𝐷 is a diagonal matrix. We do this by performing an eigen-decomposition of the Hessian in the model.
Then, we call the Galahad routine DTRS; see the Galahad3 documentation for further details.

Regularization

If model = 1, 2, or 3, and type_of_method=2, then the next step is taken to be the minimum of the model
with a regularization term added:

𝑠𝑘 = arg min
𝑠

𝑚𝑘(𝑠) +
1

∆𝑘
· 1

𝑝
‖𝑠‖𝑝𝐵 , (1.6)

At present, only one method of solving this subproblem is supported:

nlls_method = 4: this solves (1.6) by first converting the problem into the form

min
𝑝

𝑤𝑇𝑝 +
1

2
𝑝𝑇𝐷𝑝 +

1

𝑝
‖𝑝‖𝑝2,

where 𝐷 is a diagonal matrix. We do this by performing an eigen-decomposition of the Hessian in the model.
Then, we call the Galahad routine DRQS; see the Galahad3 documentation for further details.

Newton-Tensor subproblem

If model=4, then the non-quadratic Newton-Tensor model is used. As such, none of the established subproblem
solvers described in The trust region method or Regularization can be used.

If we use regularization (with 𝑝 = 2), then the subproblem we need to solve is of the form

min
𝑠

1

2

𝑚∑︁
𝑖=1

𝑊𝑖𝑖(𝑡𝑘(𝑠))𝑖
2

+
1

2∆𝑘
‖𝑠‖22 (1.7)

Note that (1.7) is a sum-of-squares, and as such can be solved by calling nlls_solve() recursively. We support
two options:

inner_method = 1 if this option is selected, then nlls_solve() is called to solve (1.4) directly. The current
regularization parameter of the ‘outer’ method is used as a base regularization in the ‘inner’ method, so that the
(quadratic) subproblem being solved in the ‘inner’ call is of the form

min
𝑠

𝑚𝑘(𝑠) +
1

2

(︂
1

∆𝑘
+

1

𝛿𝑘

)︂
‖𝑠‖2𝐵 ,

where 𝑚𝑘(𝑠) is a quadratic model of (1.4), ∆𝑘 is the (fixed) regularization parameter of the outer iteration, and
𝛿𝑘 the regularization parameter of the inner iteration, which is free to be updated as required by the method.

2 Conn, A. R., Gould, N. I., & Toint, P. L. (2000). Trust region methods. SIAM.
3 Gould, N. I., Orban, D., & Toint, P. L. (2003). GALAHAD, a library of thread-safe Fortran 90 packages for large-scale nonlinear optimization.

ACM Transactions on Mathematical Software (TOMS), 29(4), 353-372.

inner_method = 2 in this case we use nlls_solve() to solve the regularized model (1.7) directly. The
number of parameters for this subproblem is 𝑛 + 𝑚. Specifically, we have a problem of the form

min
𝑠

1

2
‖̂︀𝑟(𝑠)‖2𝑊 , where (̂︀𝑟(𝑠))𝑖 =

{︃
(𝑡𝑘(𝑠))𝑖 1 ≤ 𝑖 ≤ 𝑚

1√
Δ𝑘

𝑠𝑖 𝑚 + 1 ≤ 𝑖 ≤ 𝑛 + 𝑚
.

This subproblem can then be solved using any of the methods described in The trust region method or Regular-
ization.

inner_method = 3

In this case, nlls_solve() is called recursively with the inbuilt feature of solving a regularized prob-
lem, as described in Incorporating the regularization term

1.3.4 Accepting the step and updating the parameter

Once a step has been suggested, we must decide whether or not to accept the step, and whether the trust region radius
or regularization parameter, as appropriate, should grow, shrink, or remain the same.

These decisions are made with reference to the parameter, 𝜌 (1.2), which measures the ratio of the actual reduction
in the model to the predicted reduction in the model. If this is larger than eta_successful in nlls_options,
then the step is accepted.

The value of ∆𝑘 then needs to be updated, if appropriate. The package supports two options:

tr_update_strategy = 1 a step-function is used to decide whether or not to increase or decrease ∆𝑘, as
described here:

if 𝜌 ≤ eta_success_but_reduce then
∆ = radius_reduce *∆ // reduce Δ

else if 𝜌 ≤ eta_very_successful

∆ = ∆ // Δ stays unchanged

else if 𝜌 ≤ eta_too_successful

∆ = radius_increase *∆ // increase Δ

else if 𝜌 > eta_too_successful

∆ = ∆ // too successful: accept step, but don’t change Δ

end if

tr_update_strategy = 2 a continuous function is used to make the decision5, as described below. On the first
call, the parameter 𝜈 is set to 2.0.

if 𝜌 ≥ eta_too_successful

∆ = ∆ // Δ stays unchanged

else if 𝜌 > eta_successful

∆ = ∆ *min
(︀
radius_increase, 1−

(︀
(radius_increase− 1) * ((1− 2 * 𝜌)3)

)︀)︀
𝜈 = radius_reduce

else if 𝜌 ≤ eta_successful

∆ = 𝜈 *∆

𝜈 = 0.5 * 𝜈
end if

5 Nielsen, Hans Bruun (1999). Damping parameter in Marquadt’s MEthod. Technical report TR IMM-REP-1999-05, Department of Mathemat-
ical Modelling, Technical University of Denmark (http://www2.imm.dtu.dk/documents/ftp/tr99/tr05_99.pdf)

1.3.5 Incorporating the regularization term

If a non-zero regularization term is required in (1.1), then this is handled by transforming the problem internally into
a new non-linear least-squares problem. The formulation used will depend on the value of regularization in
nlls_options.

regularization = 1 This is only supported if p = 2. We solve a least squares problem with 𝑛 additional
degrees of freedom. The new function, ̂︀𝑟 : R𝑛 → R𝑚+𝑛, is defined as

̂︀𝑟𝑖(𝑥) =

{︃
𝑟𝑖(𝑥) for 𝑖 = 1, . . . ,𝑚
√
𝜎[𝑥]𝑗 for 𝑖 = 𝑚 + 𝑗, 𝑗 = 1, . . . , 𝑛

where [𝑥]𝑗 denotes the 𝑗th component of 𝑥.

This problem is now in the format of a standard non-linear least-squares problem. In addition to the function
values, the we also need a Jacobian and some more information about the Hessian. For our modified function,
the Jacobian is

̂︀𝐽(𝑥) =

[︂
𝐽(𝑥)√
𝜎𝐼

]︂
,

and the other function that needs to be supplied is given by

̂︁𝐻𝑘(𝑥) =

𝑛+𝑚∑︁
𝑖=1

̂︀𝑟𝑖(𝑥)∇2̂︀𝑟𝑖(𝑥) =

𝑛∑︁
𝑖=1

𝑟𝑖(𝑥)∇2𝑟𝑖(𝑥) = 𝐻𝑘(𝑥).

We solve these problems implicitly by modifing the code so that the user does not need do any additional work.
We can simply note that

‖̂︀𝑟(𝑥)‖2 = ‖𝑟(𝑥)‖2 + 𝜎‖𝑥‖2,

̂︀𝐽𝑇 ̂︀𝑟 = 𝐽𝑇𝑟 + 𝜎𝑥,

and that

̂︀𝐽𝑇 ̂︀𝐽 = 𝐽𝑇𝐽 + 𝜎𝐼.

We also need to update the value of the model. Since the Hessian vanishes, we only need to be concerned with
the Gauss-Newton model. We have that

̂︀𝑚𝐺𝑁
𝑘 (𝑠) =

1

2
‖̂︀𝑟(𝑥𝑘) + ̂︀𝐽𝑘𝑠‖2

=
1

2

(︁̂︀𝑟𝑇 ̂︀𝑟 + 2𝑠𝑇 ̂︀𝐽𝑇
𝑘 ̂︀𝑟 + 𝑠𝑇 ̂︀𝐽𝑇

𝑘
̂︀𝐽𝑘𝑠
)︁

=
1

2

(︁
𝑟𝑇𝑟 + 𝜎𝑥𝑇𝑥 + 2(𝑠𝑇𝐽𝑘

𝑇𝑟 + 𝜎𝑠𝑇𝑥) + 𝑠𝑇𝐽𝑘
𝑇𝐽𝑘𝑠 + 𝜎𝑠𝑇𝑠

)︁
= 𝑚𝐺𝑁

𝑘 (𝑠) +
1

2
𝜎(𝑥𝑇𝑥 + 2𝑠𝑇𝑥 + 𝑠𝑇𝑠)

= 𝑚𝐺𝑁
𝑘 (𝑠) +

𝜎

2
‖𝑥 + 𝑠‖2

regularization=2

We solve a non-linear least-squares problem with one additional degree of freedom.

Since the term 𝜎
𝑝 ‖𝑥‖

𝑝
2 is non-negative, we can write

𝐹𝜎(𝑥) =
1

2

⎛⎝‖𝑟(𝑥)‖2 +

(︃(︂
2𝜎

𝑝

)︂1/2

‖𝑥‖𝑝/2
)︃2
⎞⎠ ,

thereby defining a new non-linear least squares problem involving the function 𝑟 : R𝑛 → R𝑚+1 such that

𝑟𝑖(𝑥) =

{︃
𝑟𝑖(𝑥) 1 ≤ 𝑖 ≤ 𝑚
2𝜎
𝑝 ‖𝑥‖

𝑝/2 𝑖 = 𝑚 + 1
.

The Jacobian for this new function is given by

𝐽(𝑥) =

[︃
𝐽(𝑥)(︀

𝜎𝑝
2

)︀1/2 ‖𝑥‖(𝑝−4)/2𝑥𝑇

]︃
,

and we get that

∇2𝑟𝑚+1 =
(︁𝜎𝑝

2

)︁1/2
‖𝑥‖(𝑝−4)/2

(︂
𝐼 +

𝑥𝑥𝑇

‖𝑥‖2

)︂
.

As for the case where regularization=1, we simply need to update quantities in our non-linear least
squares code to solve this problem, and the changes needed in this case are

‖𝑟(𝑥)‖2 = ‖𝑟(𝑥)‖2 +
2𝜎

𝑝
‖𝑥‖𝑝,

𝐽𝑇𝑟 = 𝐽𝑇𝑟 + 𝜎‖𝑥‖𝑝−2𝑥,

𝐽𝑇𝐽 = 𝐽𝑇𝐽 +
𝜎𝑝

2
‖𝑥‖𝑝−4𝑥𝑥𝑇 ,

𝑚+1∑︁
𝑖=1

𝑟𝑖(𝑥)�̄�𝑖(𝑥) = 𝜎‖𝑥‖𝑝−4
(︀
‖𝑥‖2𝐼 + 𝑥𝑥𝑇

)︀
+

𝑚∑︁
𝑖=1

𝑟𝑖(𝑥)𝐻𝑖(𝑥)

We also need to update the model. Here we must consider the Gauss-Newton and Newton models sepa-
rately.

�̄�𝐺𝑁
𝑘 (𝑠) =

1

2
‖𝑟(𝑥𝑘) + 𝐽𝑘𝑠‖2

=
1

2

(︀
𝑟𝑇𝑟 + 2𝑠𝑇𝐽𝑇

𝑘 𝑟 + 𝑠𝑇𝐽𝑇
𝑘 𝐽𝑘𝑠

)︀
=

1

2

(︂
𝑟𝑇𝑟 +

2𝜎

𝑝
‖𝑥‖𝑝 + 2(𝑠𝑇𝐽𝑘

𝑇𝑟 + 𝜎‖𝑥‖𝑝−2𝑠𝑇𝑥) + 𝑠𝑇𝐽𝑘
𝑇𝐽𝑘𝑠 +

𝜎𝑝

2
‖𝑥‖𝑝−4(𝑠𝑇𝑥)2

)︂
= 𝑚𝐺𝑁

𝑘 (𝑠) + 𝜎

(︂
1

𝑝
‖𝑥‖𝑝 + ‖𝑥‖𝑝−2𝑠𝑇𝑥 +

𝑝

4
‖𝑥‖𝑝−4(𝑠𝑇𝑥)2

)︂
.

If we use a Newton model then

�̄�𝑁
𝑘 (𝑠) = �̄�𝐺𝑁

𝑘 (𝑠) +
1

2
𝑠𝑇𝐻𝑘𝑠

= �̄�𝐺𝑁
𝑘 (𝑠) +

1

2
𝑠𝑇
(︂
𝐻𝑘 + 𝜎‖𝑥‖𝑝−2

(︂
𝐼 +

𝑥𝑥𝑇

‖𝑥‖2

)︂)︂
𝑠

= �̄�𝐺𝑁
𝑘 (𝑠) +

1

2
𝑠𝑇𝐻𝑘𝑠 +

𝜎

2
‖𝑥‖𝑝−4𝑠𝑇

(︀
𝑥𝑇𝑥𝐼 + 𝑥𝑥𝑇

)︀
𝑠

= �̄�𝐺𝑁
𝑘 (𝑠) +

1

2
𝑠𝑇𝐻𝑘𝑠 +

𝜎

2
‖𝑥‖𝑝−4

(︀
(𝑥𝑇𝑥)(𝑠𝑇𝑠) + (𝑥𝑇𝑠)2

)︀

1.4 Examples

Consider fitting the function 𝑦(𝑡) = 𝑥1𝑒
𝑥2𝑡 to data (𝑡,𝑦) using a non-linear least squares fit.

The residual function is given by

𝑟𝑖(𝑥) = 𝑥1𝑒
𝑥2𝑡𝑖 − 𝑦𝑖.

We can calculate the Jacobian and Hessian of the residual as

∇𝑟𝑖(𝑥) =
(︀
𝑒𝑥2𝑡𝑖 𝑡𝑖𝑥1𝑒

𝑥2𝑡𝑖
)︀
,

∇2𝑟𝑖(𝑥) =

(︂
0 𝑡𝑖𝑒

𝑥2𝑡𝑖

𝑡𝑖𝑒
𝑥2𝑡𝑖 𝑥1𝑡

2
𝑖 𝑒

𝑥2𝑡𝑖

)︂
.

For some data points, 𝑦𝑖, 𝑡𝑖, (𝑖 = 1, . . . ,𝑚) the user must return

𝑟(𝑥) =

⎡⎢⎣ 𝑟1(𝑥)
...

𝑟𝑚(𝑥)

⎤⎥⎦ , 𝐽(𝑥) =

⎡⎢⎣∇𝑟1(𝑥)
...

∇𝑟𝑚(𝑥)

⎤⎥⎦ , 𝐻𝑓(𝑥) =

𝑚∑︁
𝑖=1

(𝑟)𝑖∇2𝑟𝑖(𝑥),

where, in the case of the Hessian, (𝑟)𝑖 is the 𝑖th component of a residual vector passed to the user.

𝑖 1 2 3 4 5
𝑡𝑖 1 2 4 5 8
𝑦𝑖 3 4 6 11 20

and initial guess 𝑥 = (2.5, 0.25), the following code performs the fit (with no weightings, i.e., 𝑊 = 𝐼).

! examples/Fortran/nlls_example.f90
!
! Attempts to fit the model y_i = x_1 e^(x_2 t_i)
! For parameters x_1 and x_2, and input data (t_i, y_i)
module fndef_example

use ral_nlls_double, only : params_base_type
implicit none

integer, parameter :: wp = kind(0d0)

type, extends(params_base_type) :: params_type
real(wp), dimension(:), allocatable :: t ! The m data points t_i
real(wp), dimension(:), allocatable :: y ! The m data points y_i

end type

contains
! Calculate r_i(x; t_i, y_i) = x_1 e^(x_2 * t_i) - y_i
subroutine eval_r(status, n, m, x, r, params)

integer, intent(out) :: status
integer, intent(in) :: n
integer, intent(in) :: m
real(wp), dimension(*), intent(in) :: x
real(wp), dimension(*), intent(out) :: r
class(params_base_type), intent(inout) :: params

real(wp) :: x1, x2

x1 = x(1)
x2 = x(2)
select type(params)
type is(params_type)

r(1:m) = x1 * exp(x2*params%t(:)) - params%y(:)

end select

status = 0 ! Success
end subroutine eval_r
! Calculate:
! J_i1 = e^(x_2 * t_i)
! J_i2 = t_i x_1 e^(x_2 * t_i)
subroutine eval_J(status, n, m, x, J, params)

integer, intent(out) :: status
integer, intent(in) :: n
integer, intent(in) :: m
real(wp), dimension(*), intent(in) :: x
real(wp), dimension(*), intent(out) :: J
class(params_base_type), intent(inout) :: params

real(wp) :: x1, x2

x1 = x(1)
x2 = x(2)
select type(params)
type is(params_type)

J( 1: m) = exp(x2*params%t(1:m)) ! J_i1
J(m+1:2*m) = params%t(1:m) * x1 * exp(x2*params%t(1:m))! J_i2

end select

status = 0 ! Success
end subroutine eval_J
! Calculate
! HF = sum_i r_i H_i
! Where H_i = [ 0 t_i e^(x_2 t_i) ]
! [ t_i e^(x_2 t_i) t_i^2 x_1 e^(x_2 t_i) ]
subroutine eval_HF(status, n, m, x, r, HF, params)

integer, intent(out) :: status
integer, intent(in) :: n
integer, intent(in) :: m
real(wp), dimension(*), intent(in) :: x
real(wp), dimension(*), intent(in) :: r
real(wp), dimension(*), intent(out) :: HF
class(params_base_type), intent(inout) :: params

real(wp) :: x1, x2

x1 = x(1)
x2 = x(2)
select type(params)
type is(params_type)

HF( 1) = sum(r(1:m) * 0) ! H_11
HF( 2) = sum(r(1:m) * params%t(1:m) * exp(x2*params%t(1:m))) ! H_21
HF(1*n+1) = HF(2) ! H_12
HF(1*n+2) = sum(r(1:m) * (params%t(1:m)**2) * x1 * exp(x2*params%t(1:m)))! H_

→˓22
end select

status = 0 ! Success
end subroutine eval_HF

end module fndef_example

program nlls_example

use ral_nlls_double
use fndef_example
implicit none

type(nlls_options) :: options
type(nlls_inform) :: inform

integer :: m,n
real(wp), allocatable :: x(:)
type(params_type) :: params

! Data to be fitted
m = 5
allocate(params%t(m), params%y(m))
params%t(:) = (/ 1.0, 2.0, 4.0, 5.0, 8.0 /)
params%y(:) = (/ 3.0, 4.0, 6.0, 11.0, 20.0 /)

! Call fitting routine
n = 2
allocate(x(n))
x = (/ 2.5, 0.25 /) ! Initial guess
call nlls_solve(n, m, x, eval_r, eval_J, eval_HF, params, options, inform)
if(inform%status.ne.0) then

print *, "ral_nlls() returned with error flag ", inform%status
stop

endif

! Print result
print *, "Found a local optimum at x = ", x
print *, "Took ", inform%iter, " iterations"
print *, " ", inform%f_eval, " function evaluations"
print *, " ", inform%g_eval, " gradient evaluations"
print *, " ", inform%h_eval, " hessian evaluations"

end program nlls_example

This returns the following output:

1.5 Indices and tables

• genindex

• search

eval_Hf() (fortran subroutine), 7
eval_HP() (fortran subroutine), 8
eval_J() (fortran subroutine), 6

N
nlls_inform (fortran type), 12

